

Trang chủ » Hướng dẫn lập trình module sim900A và Arduino

Hướng dẫn lập trình module sim900A và Arduino

Xin chào các bạn!

Hôm nay, mình hướng dẫn các bạn cách lập trình **module Sim900A** với Arduino Uno (Trong bài mình sử dụng mạch **Uno VN01**)

Cụ thể, bài viết hướng dẫn các bạn các phần sau:

- 1 số lệnh AT cơ bản thường dùng với Module Sim900A,
- Cách kết nối Module Sim900A với Arduino và module Relay để điều khiển bật /tắt 1 thiết bị điện
- Code tham khảo trên Arduino, để điều khiển bật/tắt thiết bị bằng cách gửi tin nhắn tới Module Sim900A

Một số lệnh AT thường dùng với module Sim900A

1.Lệnh chung

Lệnh: **AT**<CR><LF>

Trả lời: **OK**<CR><LF>

Mô tả : Kiểm tra đáp ứng của Module Sim 900A, nếu trả về OK thì Module hoạt động

Lệnh: **ATE**[x]<CR><LF>

Trả lời: **OK**<CR><LF>

Mô tả : Chế độ echo là chế độ phản hồi dữ liệu truyền đến của module Sim 900A,

x = 1 bật chế độ echo , x = 0 tắt chế độ echo (bạn nên tắt chế độ này)

Lệnh: **AT+IPR**=[baud rate]<CR><LF>

Trả lời: **OK**<CR><LF>

Mô tả : cài đặt tốc độ giao tiếp dữ liệu với module Sim 900A, chỉ cài được các tốc độ sau

baud rate : 0 (auto), 1200, 2400, 4800, 9600, 19200, 38400, 57600, 115200

2.Lệnh điều khiển cuộc gọi

Lệnh: **ATD**[Số_điện_thoại];<CR><LF>

Trả lời: **OK**<CR><LF>

Mô tả : Lệnh thực hiện cuộc gọi

Lệnh: **ATH**<CR><LF>

Trả lời: **OK**<CR><LF>

Mô tả : Lệnh thực hiện kết thúc cuộc gọi , hoặc cúp máy khi có cuộc gọi đến

Lệnh: **ATA**<CR><LF>

Trả lời: **OK**<CR><LF>

Mô tả : Lệnh thực hiện chấp nhận khi có cuộc gọi đến

Chú ý: khi nhận được cuộc gọi đến thì Module Sim 900A sẽ phản hồi về như sau

RING

RING

Nếu muốn hiển thị thông tin người gọi đến các bạn thực hiện thêm lệnh sau

KHUYẾ



340,000





R



R



mlab

1,480,000

Lệnh: **AT+CLIP=1**<CR><LF>

Trả lời: **OK**<CR><LF>

3.Lệnh điều khiển tin nhắn

Lệnh: **AT+CMGF=1**<CR><LF>

Trả lời: **OK**<CR><LF>

Mô tả : Lệnh đưa SMS về chế độ Text , phải có lệnh này mới gửi nhận tin nhắn dạng Text

Cấu trúc gửi tin nhắn

Lệnh: **AT+CMGS="Số_điện_thoại"**<CR><LF>

Đợi đến khi có ký tự '>' được gửi về

Gửi nội dung tin nhắn : **"This is a test"**

Gửi mã Ctrl+Z hay **0x1A** để kết thúc nội dung và gửi tin nhắn

Trả lời : **OK**<CR><LF>

Lệnh: **AT+CMGR=1**<CR><LF>

Mô tả : Đọc một nhắn vừa gửi đến, lệnh được trả về nội dung tin nhắn, thông tin người gửi, thời gian gửi

Chú ý: Nếu Module Sim 900A nhận được một tin nhắn bất kỳ, thì nó gửi về cụm **"+CMTI"** để thông báo

Trong trường hợp bạn muốn hiển thị trực tiếp nội dung tin nhắn(không lưu vào bộ nhớ của sim) bạn gửi lệnh sau

AT+CNMI=2,2<CR><LF>

Chú ý: sau mỗi lệnh các bạn thường thấy <CR><LF> thực chất nó là hai mã điều khiển <CR> tương ứng 0x0D(hexa)

<LF> tương ứng 0x0A(hexa)

nếu các bạn test trên máy tính sau mỗi lệnh các bạn chỉ cần ấn phím Enter

Ví dụ : gõ AT rồi ấn Enter

Chuẩn bị phần cứng và kết nối

1 Nguồn 12VDC/1A

<http://mlab.vn/1272384-adapter-dc-12v-1a.html>

1 Sim900A Shield

<http://mlab.vn/408321-module-sim900a.html>

1 Uno VN01

<http://mlab.vn/1697809-uno-vn01-arduino-uno-phiên-bản-việt-nam.html>


1 Module Relay 12V

<http://mlab.vn/186624-module-relay-12vdc.html>

1 thiết bị điện mà bạn muốn điều khiển. Trong bài hướng dẫn này mình sử dụng bóng đèn 12V.

Dây kết nối

Bạn có thể xem hình sẽ thấy trực quan hơn



Quy trình kết nối

Đơn giản, chỉ cần cắm Shield Sim900A lên kit Arduino Uno



Lúc này ác dây chân Gnd, Rx, Tx, PWR_KEY được kết nối giữa Module Sim900A và Arduino

Tiếp theo dùng dây để kết nối các thành phần còn lại (bạn kết nối như hình mô tả trên cùng)



Cắm dây USB 5V vào Arduino , Cắm nguồn 12V vào Shield Sim900A để cấp nguồn cho module Sim900A hoạt động

Như vậy ta đã chuẩn bị xong phần cứng

Phần lập trình , các bạn mở Arduino IDE lên và copy đoạn code dưới vào

Trong code mình đã ghi chú khá chi tiết

```
// Bai toan : nhan tin dieu khien bat tat bong den
// Neu noi dung tin nhan la : LAMP_ON      thi bat bong den
// Neu noi dung tin nhan la : LAMP_OFF     thi tat bong den
```

```
const String myphone = "01689951815"; // Thay so cua ban vao day
const int PWR_KEY = 9; // Chan so 9 arduino uno dung lam chan dieu khien bat tat module sim900A
const int RELAY = 12; // Chan so 12 arduino uno dung lam chan dieu khien dong/cat Relay de On/Off dem
```

```
String RxBuff = ""; // Khai bao bo dem nhan du lieu
// Tat ca du lieu nhan ve tu module sim deu duoc luu trong day
```

```
int Index_Lamp_On = -1;           // vi tri cua chuoi "LAMP_ON"
int Index_Lamp_Off = -1;         // vi tri cua chuoi "LAMP_OFF"
```

```
void Gsm_Power_On();           // Bat module Sim 900A
void Gsm_Init();               // Cau hinh Module Sim 900A
void Gsm_MakeCall(String phone); // Ham goi dien
void Gsm_MakeSMS(String phone,String content); // Ham nhan tin
```

```

void setup() {
    Serial.begin(9600);           // Cấu hình UART để giao tiếp module Sim 900A

    digitalWrite(RELAY, LOW);     // Khai báo chân để điều khiển đóng cắt RELAY
    pinMode(RELAY, OUTPUT);

    digitalWrite(PWR_KEY, LOW);   // Khai báo chân PWR_KEY để điều khiển bật tắt module Sim 900A
    pinMode(PWR_KEY, OUTPUT);

    delay(1000);
    Gsm_Power_On();              // Bật Module Sim 900A
    delay(10000);
    Gsm_Init();                  // Cấu hình module Sim 900A
    Gsm_MakeCall(myphone);       // Test cuộc gọi
    Gsm_MakeSMS(myphone, "I'm a test"); // Test tin nhắn
}

void loop() {
    delay(1000);                 // trễ 1s
    Index_Lamp_On = RxBuff.indexOf("LAMP_ON"); // Tìm vị trí của chuỗi "LAMP_ON" trong bộ đệm nhận RxBuff
    if(Index_Lamp_On >= 0)       // Nếu tìm thấy "LAMP_ON" trong RxBuff
    {
        Index_Lamp_On = -1;      //
        RxBuff = "";             // Xóa bộ đệm
        digitalWrite(RELAY, HIGH); // Đóng Relay để bật đèn // Bật bóng đèn
    }
    else
    {
        Index_Lamp_Off = RxBuff.indexOf("LAMP_OFF"); // Nếu không tìm thấy "LAMP_ON" thì tiếp tục tìm "LAMP_OFF"
        if(Index_Lamp_Off >= 0) // Tìm vị trí của chuỗi "LAMP_OFF" trong bộ đệm nhận RxBuff
        {
            Index_Lamp_Off = -1; //
            RxBuff = "";         // Xóa bộ đệm
            digitalWrite(RELAY, LOW); // Cắt Relay để tắt đèn
        }
    }
}

void serialEvent() {
    while (Serial.available()) { // Chương trình ngắt nhận dữ liệu
        // Đọc đến khi có dữ liệu nhận về
        // get the new byte:
        char inChar = (char)Serial.read(); // Đọc một byte dữ liệu vừa nhận về
        RxBuff += inChar;                 // Ghi byte đó vào bộ đệm nhận RxBuff (ta sẽ xử lý RxBuff trong vòng loop())
        if(RxBuff.length() >= 128)       // Nếu bộ đệm quá dài thì xóa bộ đệm đi
        {
            RxBuff = "";
        }
    }
}

void Gsm_Power_On()
{
    digitalWrite(PWR_KEY, HIGH); // Đặt chân PWR_KEY lên cao ít nhất 1s
    delay(1500);                 // ở đây ta để 1,5s
    digitalWrite(PWR_KEY, LOW);  // Đặt chân PWR_KEY xuống thấp
    delay(100);                  // các bạn xem trong Hardware designed sim900A để hiểu rõ hơn
}

void Gsm_Init()
{
    Serial.println("ATE0");      // Tắt chế độ phản hồi (Echo mode)
    delay(2000);
    Serial.println("AT+IPR=9600"); // Đặt tốc độ truyền nhận dữ liệu 9600 bps
    delay(2000);
    Serial.println("AT+CMGF=1"); // Chọn chế độ TEXT Mode
    delay(2000);
    Serial.println("AT+CLIP=1"); // Hiện thị thông tin người gọi đến
    delay(2000);
    Serial.println("AT+CNMI=2,2"); // Hiện thị trực tiếp nội dung tin nhắn
    delay(2000);
}

void Gsm_MakeCall(String phone)
{
    Serial.println("ATD" + phone + ";"); // Gọi điện
    delay(10000);                         // Sau 10s
    Serial.println("ATH");                 // Ngắt cuộc gọi
    delay(2000);
}

void Gsm_MakeSMS(String phone, String content)
{
    Serial.println("AT+CMGS=\"" + phone + "\""); // Lệnh gửi tin nhắn
    delay(3000);                                // Cho ký tự '>' phản hồi về
    Serial.print(content);                      // Gửi nội dung
    Serial.print((char)26);                    // Gửi Ctrl+Z hay 26 để kết thúc nội dung tin nhắn và gửi tin đi
    delay(5000);                               // delay 5s
}

```

Các ấn biên dịch rồi nạp chương trình xuống

Chú ý: Do Arduino Uno R3 chỉ có 1 cổng UART, mà lại có hai chức năng

- Nạp chương trình xuống Chip
- Giao tiếp Module Sim900A

Vì thế trong quá trình nạp bạn cần tắt Module Sim900A đi (trên module có một nút ấn, bạn ấn giữ khoảng 2 s rồi nhả ra)

Thấy đèn trên module không nhấp nháy nữa thì bạn mới nạp chương trình xuống


Trong chương trình: có lệnh test gọi điện và gửi tin nhắn ngay khi bắt đầu chạy.

Dựa vào điều này bạn lấy số trên của Sim và thực hiện điều khiển bật tắt bóng đèn qua điện thoại

Cú pháp **"LAMP_ON"** bật đèn

LAMP_OFF" tắt đèn

Và đây là kết quả



[Mlab] Hướng dẫn lập trình module sim900A và Arduino



Chúc các bạn thành công!

phan quang | ★★★★★
06/25/2020

em đang muốn lập trình điều khiển tác động đóng ngắt hệ thống điện trong nhà mà dùng arduino nano và sim900a nhưng lập trình theo cái này nó ko chạy mong a giúp

NGO HUNG | ★★★★★
01/19/2019

CHƯƠNG TRÌNH KHÔNG HỀ NHẮC TỚI HÀM serialEvent() LẤY GI MÀ HỆ THỐNG HIẾU VẬY

Trần Bình Minh | ★★★★★
09/01/2018

Làm sao để báo trạng thái của thiết bị về điện thoại vậy a?

PHAN VAN HAI | 
04/06/2018

ANH CHO EM HOI LA EM MUON GUI NHiet DO TU CAM BIEN QUA SMS VE DIEN THOI THI GUI NHU THE NAO A

Ngo Van Hien | 
01/17/2018

MLAB hay!!!!!!!!!!!!!!!!!!!!

Viết đánh giá

Họ và tên:

Đánh giá của bạn:

Lưu ý: Không hỗ trợ HTML!

Bình chọn: ☐ Dở ☐ ☐ ☐ ☐ Hay

Nhập mã bảo vệ:

d9fc6b

Tiếp tục